Всем здравствуйте.
Имеется простенький запрос:

```1c
    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Контрагент,
    | ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Договор,
    | ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт КАК НачальныйОстаток,
    | ХозрасчетныйОстаткиИОбороты.СуммаОборотДт КАК Выдано,
    | ХозрасчетныйОстаткиИОбороты.СуммаОборотКт КАК Возвращено,
    | ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт КАК КонечныйОстаток
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачПериода, &КонПериода, Регистратор, , Счет = &СчетЗаймы, , ) КАК ХозрасчетныйОстаткиИОбороты
    |ИТОГИ
    | СУММА(НачальныйОстаток),
    | СУММА(Выдано),
    | СУММА(Возвращено),
    | СУММА(КонечныйОстаток)
    |ПО
    | ОБЩИЕ,
    | Контрагент,
    | Договор";

    Запрос.УстановитьПараметр("КонПериода", КонПериода);
    Запрос.УстановитьПараметр("НачПериода", НачПериода);
    Запрос.УстановитьПараметр("СчетЗаймы", ПланыСчетов.Хозрасчетный.НайтиПоКоду("58.03"));
```

Ставлю период сентябрь 2020 года. Все хорошо и прекрасно работает. Остатков на начало нет, есть обороты и конечные остатки. Все совпадает с ОСВ по счету.

Помещаю запрос во временную таблицу, чтобы сделать еще один запрос по другому счету и совместить.

```1c
    Запрос = Новый Запрос;
    Запрос.Текст =
    "ВЫБРАТЬ
    | ХозрасчетныйОстаткиИОбороты.Субконто1 КАК Контрагент,
    | ХозрасчетныйОстаткиИОбороты.Субконто2 КАК Договор,
    | ХозрасчетныйОстаткиИОбороты.СуммаНачальныйОстатокДт КАК НачальныйОстаток,
    | ХозрасчетныйОстаткиИОбороты.СуммаОборотДт КАК Выдано,
    | ХозрасчетныйОстаткиИОбороты.СуммаОборотКт КАК Возвращено,
    | ХозрасчетныйОстаткиИОбороты.СуммаКонечныйОстатокДт КАК КонечныйОстаток
    |ПОМЕСТИТЬ Займы
    |ИЗ
    | РегистрБухгалтерии.Хозрасчетный.ОстаткиИОбороты(&НачПериода, &КонПериода, Регистратор, , Счет = &СчетЗаймы, , ) КАК ХозрасчетныйОстаткиИОбороты
    |;
    |
    |////////////////////////////////////////////////////////////­////////////////////
    |ВЫБРАТЬ
    | Займы.Контрагент КАК Контрагент,
    | Займы.Договор КАК Договор,
    | Займы.НачальныйОстаток КАК НачальныйОстаток,
    | Займы.Выдано КАК Выдано,
    | Займы.Возвращено КАК Возвращено,
    | Займы.КонечныйОстаток КАК КонечныйОстаток
    |ИЗ
    | Займы КАК Займы
    |ИТОГИ
    | СУММА(НачальныйОстаток),
    | СУММА(Выдано),
    | СУММА(Возвращено),
    | СУММА(КонечныйОстаток)
    |ПО
    | ОБЩИЕ,
    | Контрагент,
    | Договор";
```

Период формирования прежний - сентябрь 2020 года.

И тут же появляются остатки на начало равные оборотам. Причем, сами обороты тоже остаются.

В чем может быть причина такого формирования? Проверено на трех базах, везде одинаковое поведение.
---
1. Ну начнем с того, что периодичность по регистратору тут вообще не нужна.
2. В первом запросе у вас итоги указаны по договору и контрагентам, у вас практически нет детальных записей по регистраторам - все они корректно считаются при выводе итогов (без сворачивания).
3. Во втором запросе у вас формируется таблица записей по регистраторам в хронологии (абсолютно все детальные записи). Естественно, если по договору и контрагенты было несколько документов, то на каждый второй и последующий регистраторы - появляются начальные остатки (являющиеся конечным остатком для предыдущего регистратора). То есть у вас в ВТ Займы получается типа ОСВ.
4. Ну и потом вы тупо сворачиваете эту ОСВ по колонкам. Представьте просто, что будет если ОСВ вывести в Ёксель и под всеми колонками подвести итог по сумме... Это будет правильный итог?

Кстати, что там с конечным остатком во втором запросе? ;-)
---
Как вы определили, что нет Начальных остатков в первом случае?
Все есть и в первом и во втором случае.
---
Я подозреваю, что там просто до сентября просто не было хозяйственной деятельности...
---
исходя из слов ТС, СУБД "дорисовывает" начальные остатки = оборотам во временную таблицу.
Магия 5-го уровня прям.
---
Не, она дорисовывает из ВТ. На самом деле я вижу все по другому, вчера в комменте (3) написал...
---
Так запрос простейший, без группировок и прочей фигни, его результат - без начальных остатков.
Простейшая выборка из ВТ уже начинает содержать начальные остатки.

Я читал (3) - не согласен, т.к. ОстаткиИОбороты выдает результат уже в "свернутом" виде

Пусть даже указана детализация до регистратора, но он не выводится в поля - все свернется до тех полей, которые выводятся в результате.

Чую, ТС просто периоды разные использует(следующий месяц от первого случая), поэтому у него во втором случае появляются начальные остатки.
---
Ну давайте все вместе попросим ТС показать промежуточную ВТ Займы...
---
Джентельменам верят на слово © анекдот :)
---
Не верю! (С) Станиславский. )))
---
Да
---
мб поменять ", Регистратор," на ",Авто,"
---
1. Ну начнем с того, что периодичность по регистратору тут вообще не нужна.
2. В первом запросе у вас итоги указаны по договору и контрагентам, у вас практически нет детальных записей по регистраторам - все они корректно считаются при выводе итогов (без сворачивания).
3. Во втором запросе у вас формируется таблица записей по регистраторам в хронологии (абсолютно все детальные записи). Естественно, если по договору и контрагенты было несколько документов, то на каждый второй и последующий регистраторы - появляются начальные остатки (являющиеся конечным остатком для предыдущего регистратора). То есть у вас в ВТ Займы получается типа ОСВ.
4. Ну и потом вы тупо сворачиваете эту ОСВ по колонкам. Представьте просто, что будет если ОСВ вывести в Ёксель и под всеми колонками подвести итог по сумме... Это будет правильный итог?

Кстати, что там с конечным остатком во втором запросе? ;-)
---
Спасибо вам
Убрала регистратор и всё встало на свои места